

## SELECT IF

SELECT IF (logical expression)

### Relational Operators:

EQ or =	Equal to	NE or ~= or <>	Not equal to
LT or <	Less than	LE or <=	Less than or equal to
GT or >	Greater than	GE or >=	Greater than or equal to

### Logical Operators:

AND or & OR or | NOT or ~

### Missing-Value Functions:

SYSMIS Returns 1 (true) if value is system-missing  
MISSING Returns 1 (true) if value is system- or user-missing  
VALUE Includes all values except system-missing

### Example:

```
SELECT IF (SEX EQ 'MALE ' )  
SELECT IF (AVAR GE BVAR).  
SELECT IF (SYSMIS(VARA)).
```

### Overview

The SELECT IF transformation permanently selects cases for analysis based upon logical conditions found in the data. These conditions are specified in a *logical expression*. The logical expression can contain relational operators, logical operators, missing-value functions, and arithmetic operations and functions allowed in COMPUTE transformations (see COMPUTE, Universals: Logical Expressions, and Universals: Functions). For temporary case selection, see PROCESS IF.

### Syntax

- The logical expression must be enclosed in parentheses.
- Parentheses also may be used to specify the order of evaluation.
- Relational operators are EQ, NE, LT, LE, GT, and GE (and their symbolic equivalents).
- Each relational operator must be preceded by a variable name.
- Logical operators are AND, OR, and NOT (and their symbolic equivalents).
- At least one relation, SYSMIS function, or MISSING function must be included in the logical expression.
- A relation includes a variable name, the relational operator, and a value or variable.
- Long string variables cannot be used in the SELECT IF command.
- SELECT IF can be entered anywhere in an SPSS/PC job, except between BEGIN DATA and END DATA.

### Operations

- SELECT IF is a transformation and is executed when the data are read for the next procedure.
- SELECT IF permanently selects cases.
- The logical expression is evaluated as true or false.
- If a logical expression is true, the case is selected; if it is false or missing, the case is not selected.
- Multiple SELECT IF commands issued prior to a procedure command must all be true for a case to be selected.
- SELECT IF should be placed before other transformations for efficiency considerations.
- Logical expressions are evaluated in the following order: first numeric functions, then exponentiation, then arithmetic operators, then relational operators, and finally logical operators.
- Use parentheses to change the order of evaluation.

**Limitations** • The complexity of logical expressions is limited by available memory.

**Example** `SELECT IF (SEX EQ 'MALE ').`

- All subsequent procedures will use only cases in which the value of SEX is equal to MALE.
- SEX is a short string variable with a format width of six characters. The specification for value MALE includes trailing blanks to fill the field.

**Example** `SELECT IF (INCOME GT 75000 OR INCOME LE 10000).`

- The logical expression tests whether a case has a value greater than 75000 or less than 10000. If either relation is true, the case is used in subsequent analyses. Note that each relational operator (GT and LE) is preceded by a variable name.

**Example** `SELECT IF (AVAR GE BVAR).`

- This example selects cases where variable AVAR is greater than or equal to BVAR.

`SELECT IF (SEX = 'F' & INCOME <= 10000).`

- The logical expression tests whether short string variable SEX is equal to F and if numeric variable INCOME is less than or equal to 10000. Cases that meet both conditions are included in subsequent analyses.

**Example** `SELECT IF (SYSMIS(VARA)).`

- The logical expression tests whether VARA is equal to the system-missing value. If the value of VARA is system-missing, the case is selected for subsequent analyses.

**Example** `SELECT IF (VALUE(VARA) GT 0).`

- The logical expression determines if any values, including user-missing values, are greater than 0. When the expression is true, cases are selected for analysis.

`SEL IF (RECEIV GT DUE OR (REVNUS GE EXPNS AND BALNCE GT 0)).`

- The expression uses parentheses to change the order of evaluation. First, SPSS/PC tests whether variable REVNUS is greater than or equal to variable EXPNS and variable BALNCE has a value greater than 0. Second, SPSS/PC tests whether RECEIV is greater than DUE. If either of these conditions are met, the case is included in the active file for subsequent analyses.

**Example** `SELECT IF ((V1-15) LE (V2*(-0.001))).`

- The logical expression compares whether V1 minus 15 is less than or equal to V2 multiplied by -0.001. If the expression is true, the case is selected for subsequent analyses.

**Example** `SELECT IF ((YRMODA(84,13,0) - YRMODA(YVAR,MVAR,DVAR)) LE 30).`

- The logical expression subtracts the number of days representing the date YVAR (year), MVAR (month), and DVAR (day) from the number of days representing the last day in 1984. If the difference is less than or equal to 30, the case is selected for subsequent analyses.

# SET

```
SET [SCREEN={ON }]/ [DISK={OFF  
                  {ON  
                  'filename' } }]/ [PRINTER={OFF }]/  
  
          [LENGTH={24 }]/          [WIDTH={79  
                  {n }                  132  
                                      {n  
                                      NARROW  
                                      WIDE } }]/ [EJECT={OFF }]/  
  
          [BOXSTRING={'-|+' }]/ [HISTOGRAM={'█'  
                                  {'char' } }]/ [BLOCK={'█'  
                                  {'char' } }]  
  
          [LOG={ON  
              {OFF  
              'filename' } }]/ [RESULTS={'SPSS.PRC' }]/  
                                          {'filename' }]  
  
          [ECHO={OFF }]/ [INCLUDE={ON }]  
                          {ON }                  {OFF }  
  
          [PROMPT={'SPSS/PC: ' }]/ [CPROMPT={'  
                          {'string' } }]  
  
          [ENDCMD={'.' }]/ [BEEP={ON }]/  
                          {'a' }                  {OFF }  
  
          [SEED={RANDOM }]/ [BLANKS={'.'  
                          {number }              {real num } }]
```

When SCREEN is OFF, the following defaults are in effect:

```
LENGTH=59  
EJECT=ON  
BOXSTRING='-|+'  
HISTOGRAM='X'  
BLOCK='X'
```

## Example:

```
SET DISK=ON/ECHO=ON/LENGTH=59/EJECT=ON.
```

## Overview

The SET command changes SPSS/PC running options. To see the current running options, use the SHOW command.

### Defaults

SPSS/PC directs the output from statistical and reporting procedures to your screen, with a length of 24 rows and a width of 79 columns. The default running options include command prompt SPSS/PC: , the continuation prompt ;, and the period (.) as the command terminator. Any commands you enter are sent to a log file. By default, this file is SPSS.LOG. Commands processed from a command file with the INCLUDE command are shown on the screen. If you write out a data set or matrix materials, SPSS/PC writes the results to the file SPSS.PRC.

Crosstabulation tables use special characters on the screen. Histograms (available in FREQUENCIES) use a solid bar. Icicle plots (available in CLUSTER) use a solid box. The SAMPLE transformation and UNIFORM and NORMAL functions (available in COMPUTE) use a random seed obtained from the clock. Numeric blanks are interpreted as system-missing values and printed as a period (.) in output. To remind you to move to the next screen of output, a high-pitched beep is sounded. Error messages are accompanied by a low-pitched beep.

### Tailoring

**Output Destination.** You can send your output to multiple destinations (screen, disk, and printer). You can change destinations throughout the course of an SPSS/PC session. You can direct output to the default disk file SPSS.LIS in the current directory or to a file of your choice.

**Output Form and Layout.** You can change the length and width of output. Additionally, include printer carriage-control characters for page ejects in disk file output.

**Special Characters.** You can change the characters SPSS/PC uses to print crosstabulation table grids. You can also specify the symbols SPSS/PC uses in histograms (available in FREQUENCIES) and in icicle plot symbols (available in CLUSTER).

**Optional Output.** SPSS/PC automatically saves all the commands from your SPSS/PC session in the file SPSS.LOG in the current directory. The log can be turned off or directed to a file of your choice. Log files can be used as command files in subsequent SPSS/PC sessions (see INCLUDE).

You can write data from the WRITE procedure or matrix materials from REGRESSION, CORRELATION, CLUSTER, ONEWAY, and FACTOR procedures to either SPSS.PRC or to a file of your choice.

**Command Printback.** You can direct all your SPSS/PC commands to your output file and or printer. You can also suppress the screen printback of commands included with the INCLUDE command.

**Prompts.** You can specify your own command prompt and continuation prompt. You can also change the default terminator from a period to a character of your choice and can suppress the beeps for paging and error messages.

**Random Number Seed.** You can change the initial seed value to a particular number or have SPSS/PC select a random number.

**Treatment of Numeric Blanks.** You can substitute a numeric value for blank fields in numeric variables. All numeric blanks encountered are translated to this value.

## Syntax

- Subcommands on SET can be named in any order.
- Subcommands can be separated by an optional slash.
- Only one keyword or argument can be specified for each subcommand.
- SET can be used more than once in an SPSS/PC session.
- Each time SET is entered, only the named specifications are affected. All others remain at their previous settings or the default.
- The alias for keyword ON is YES.
- The alias for keyword OFF is NO.
- The log and results files can be directed only to the current directory.

## Operations

- SET is an operation command and takes effect immediately.
- By default, SET running options assume screen output priority, using characters that may not be available on your printer.
- By default, all SPSS/PC output is directed to the screen as commands are processed.
- All files are built in the current directory.
- SPSS.LIS and SPSS.LOG are erased at the *beginning* of each SPSS/PC session. These files should be renamed (using the DOS *RENAME* command) if you want to save the contents of the files.
- If you use multiple disk files, you should not start with one file, go to another file, and then return to the first file. SPSS/PC writes over the first set of results and does *not* append.

## Limitations

- Some characters used in screen displays are not available on all printers. Additionally, some of the characters used for the screen do not have a printable equivalent.

**Example** SET DISK=ON/ECHO=ON/LENGTH=59/EJECT=ON.

- This example requests that SPSS/PC send output to the default disk file SPSS.LIS as well as to the screen. SCREEN is not turned off.
- The specification ECHO=ON integrates all input commands in the output file.
- SPSS/PC uses a page length of 59 for both screen and disk output. A MORE prompt appears in the status area after SPSS/PC produces 59 lines of output from procedures.
- EJECT=ON inserts a page-eject carriage-control character in the disk output file after each 59 lines of output and suppresses the dashed line between pages.

## Output Destination

By default, SPSS/PC assumes that output is directed only to the screen. Additionally, you can direct output directly to a printer and/or a disk file.

- When SCREEN=ON is in effect, the output is, by default, formatted especially for screen-by-screen displays and includes special graphics characters that are available only on the screen. If you turn screen display off, SPSS/PC uses different defaults for output. The defaults for SCREEN=ON and SCREEN=OFF are listed in the table below. You can change these options with other SET subcommands.
- When you direct output to both the screen and printer or disk file, you must specify format options that are compatible with your printer.
- The output from different procedures can be directed to different disk files by naming a different file on the DISK subcommand prior to a procedure command. When you use multiple disk files, you cannot send output to one file, then to another file, and then resume sending output to the first file. If you do, SPSS/PC writes over the first set of output. It cannot append output to a file once another file has been named.

### Defaults for SCREEN settings

SCREEN	LENGTH	BOXSTRNG	HISTOGRAM	BLOCK	EJECT
ON	24	- +	■	■	OFF
OFF	59	- +	X	X	ON

**SCREEN** *Screen output.* SCREEN=ON is the default. You can suppress screen display by specifying SCREEN=OFF and then restart screen display by specifying SCREEN=ON.

**PRINTER** *Printer output.* With PRINTER=ON, output is sent directly to your printer. Your printer must be turned on if you specify PRINTER=ON. You can stop sending output to the printer by specifying PRINTER=OFF.

**DISK** *Disk output.* You can direct your output to a disk file in addition to (or instead of) the screen by specifying DISK=ON. The default disk output file is SPSS.LIS in the current directory. You can direct the output to another disk file in a directory of your choice by specifying a filename on the DISK subcommand. To stop sending output to disk, specify DISK=OFF.

## Example

```
SET DISK=ON.
GET FILE='MARCH.SYS'.
LIST VAR=V1 TO V10.
SET DISK='MEANS.LIS'
MEANS V1 TO V5 BY V7.
SET DISK=OFF.
FREQ VAR=V1, V3, V4/HISTOGRAM.
SET DISK=ON/HISTOGRAM='X'.
FREQ VAR=V3/HISTOGRAM.
FINISH.
```

- The first SET command directs output to the default file SPSS.LIS in the current directory.

- The GET command reads the SPSS/PC system file MARCH.SYS from the current directory.
- The LIST procedure output is directed to the screen and to SPSS.LIS.
- The second SET command directs output to the file MEANS.LIS in the current directory.
- The output from the MEANS procedure is directed to MEANS.LIS and to the screen.
- The next SET command suspends sending output to the file MEANS.LIS. Procedure display continues coming to the screen.
- The FREQUENCIES command requests frequency tables and histograms for three variables. The histograms use the default histogram characters for the screen.
- The next SET command with DISK=ON resumes sending output to the file MEANS.LIS. The character used in any histogram on the screen and in the disk file is X.
- The FREQUENCIES command produces a table and histogram for the variable V3. Output is directed to both the screen and to the disk file.
- The FINISH command ends the SPSS/PC session.
- The file SPSS.LIS contains the results from the LIST procedure. The file MEANS.LIS contains output from the MEANS procedure and from the last FREQUENCIES procedure.

## Output Form and Layout

SCREEN=ON and SCREEN=OFF set up different defaults for output format (see table). You can change these defaults with the LENGTH, WIDTH, and EJECT subcommands.

- The LENGTH subcommand affects the screen, printer, and disk file page length.
- On the screen, the MORE prompt appears after the specified number of lines have been displayed.
- For printer and disk output, a dotted line is printed or a page eject occurs at the end of the specified number of lines.
- The WIDTH subcommand affects screen, printer, and disk output.
- The screen allows a width of only 79 characters. Lines longer than 79 characters are wrapped onto two lines.
- Printer and disk output can contain lines longer than 79 characters.
- To print very long lines, use the DOS MODE command to reduce the size of characters on your printer.
- The EJECT subcommand affects only printer and disk file output. You can print output that skips over the paper page perforations. With LENGTH=59 and EJECT=ON, insert printer paper one-half inch above the type head; SPSS/PC output will print each page with skips over subsequent perforations.

**LENGTH** *The page length.* Specify an integer number. The screen takes up 24 lines, so this is the recommended length for screen output. A standard printer page length is 59 lines.

**WIDTH** *The page width.* Specify an integer or keywords NARROW (79 characters) or WIDE (130 characters). Most SPSS/PC output is formatted to fit in a page width of 79 characters. For procedures that let you specify your own width, the width specified in the procedure should agree with the width specified on SET.

**EJECT** *Carriage control for page ejects for printer and disk output.* Specify ON or OFF. With EJECT=OFF output sent to the printer and/or to a disk file contains a dotted line indicating page breaks. With EJECT=ON, no dotted line is printed, and pages are printed with margins between each page. When SCREEN=ON, the default is EJECT=OFF.

**Example** SET DISK=ON/LENGTH=59/WIDTH=130/EJECT=ON.  
 REG VAR=SAVINGS POP15 POP75 INCOME GROWTH  
 /SELECT IN2 NE 1  
 /DEP SAVINGS /ENTER POP15/RES SEPARATE  
 /CASEWISE DEP OUTLIER (3) ALL PLOT (RESID)  
 /DEP SAVINGS /ENTER INCOME/RES POOLED.

- The SET command directs output to the default file SPSS.LIS in addition to the screen. The output is formatted with 59 lines for each page with an eject after each page. The width of each page is 130 characters.
- The REGRESSION procedure uses 130 characters on each line of the output display. On the screen, lines longer than 79 characters are wrapped.
- If SPSS.LIS is printed, each page includes top and bottom margins to skip over the perforations. No lines are wrapped.

## Special Characters

SPSS/PC uses special graphic characters in tables and plots. These graphic characters appear on the screen but may not print correctly (depending upon the type of printer you are using) or may be interpreted as special printer escape sequences. Three subcommands, BOXSTRING, HISTOGRAM, and BLOCK, are used to specify other characters.

**BOXSTRING** *Box-building characters.* Specify the character for the horizontal bar, vertical bar, and intersection, in that order and enclosed in apostrophes. If the display is going to the screen, the defaults are single-stroke box characters (ASCII decimal 196, 179, and 197). If the display is not directed to the screen, the defaults are -|-.

**HISTOGRAM** *The character used in histograms and barcharts.* Specify a single character in apostrophes. If output is directed to the screen, the default is a lower-half solid block (ASCII decimal 220). If output is not directed to the screen, the default is X.

**BLOCK** *The character used in icicle plots.* Specify a single character in apostrophes. If output is directed to the screen, the default is a full solid block (ASCII decimal 219). If output is not directed to the screen, the default is X.

You can enter special characters by pressing the "Alt" key and entering the ASCII decimal value for the character desired on the numeric keypad on the right side of the keyboard. For example, to enter the lightest-shade block character, depress the Alt key and hold it down while entering 176. Depending upon the editor you use to build command files, you may be able to enter them in the same manner.

A complete list of characters and their ASCII decimal values is available in the DOS BASIC manual. The table below documents some of the more useful characters and their ASCII decimal codes.

### Special characters

Character	ASCII value
Single stroke horizontal bar	196
Single stroke vertical bar	179
Single stroke intersection	197
Double stroke horizontal bar	205
Double stroke vertical bar	186
Double stroke intersection	206
Light intensity block	176
Medium intensity block	177
Heavy intensity block	178
Solid block	219
Lower-half solid block	220
Upper-half solid block	223
Left-half solid block	221
Right-half solid block	222
Small solid block	254

## Optional Output

SPSS/PC provides two types of additional output: a log file, which includes all commands processed in a session; and a results file, which contains casewise or matrix data produced as the result of SPSS/PC procedures.

### LOG Subcommand

By default, SPSS/PC copies all the commands you enter and that are processed into log file SPSS.LOG in the current directory. Use the LOG subcommand to change these defaults.

- You can suppress the building of the log file by specifying LOG=OFF.
- You can send the log file to a file other than SPSS.LOG by specifying a different filename on the LOG subcommand.
- The log file contains each command you enter during a session.
- If you enter a command with errors, the log file contains a comment line after that command indicating the error.
- The log file can be edited to remove erroneous lines for use as a command file (see INCLUDE).
- When you use a log file as a command file, you must initially send the log file to a file other than SPSS.LOG or rename the log file using the DOS *RENAME* command.

When you use the INCLUDE in your session, the commands from the included command file are preceded by a square bracket ([]) in the log file. If this log file is then used as a command file in a subsequent session, only the INCLUDE command is processed, not the commands preceded by the square bracket. This avoids double processing of commands. You cannot enter the square bracket yourself in interactive sessions.

### Example

```
SET LOG='DATADEF.FIL'.
DATA LIST FILE='MARCH.DAT'/OP1 TO OP7 1-14.
VARIABLE LABELS OP1 'Opinion on sex'/OP2 'Opinion on religion'/
                OP7 'Opinion on divorce'.
VALUE LABELS OP1 TO OP7 1 'Agree' 2 'Disagree'.
MISSING VALUE OP1 TO OP7(9).
INCLUDE 'DES.INC'.
FIN.
```

- The SET command directs the log file to DATADEF.FIL in the current directory.
- The DATA LIST, VARIABLE LABELS, VALUE LABELS, MISSING VALUE, and INCLUDE commands and their specifications are copied to the file. Each of the commands processed from the INCLUDE file, DES.INC, are copied to the log file prefaced with a square bracket.
- The file DATADEF.FIL can be used as a command file in a subsequent SPSS/PC session.

### RESULTS Subcommand

Some SPSS/PC procedures produce data that can be used in other sessions. WRITE produces a rectangular data file of individual values. Procedures CLUSTER, CORRELATION, FACTOR, ONEWAY, and REGRESSION produce summary data in the form of matrix materials (see individual commands). SPSS/PC writes these data files to the default file SPSS.PRC in the current directory. You can direct these data files to other files in the current directory by specifying a filename in apostrophes on the RESULTS subcommand. The results file can be used as input (named on the DATA LIST or INCLUDE command) in a subsequent SPSS/PC session.

### Example

```
SET DISK ON/RESULTS='NEWDAT.DAT'.
GET FILE='INVENT.SYS'.
COMPUTE TOTCOST=V1+V7+V9.
SELECT IF (V3 LT 7).
WRITE VARIABLES=ID TO V15, TOTCOST.
FINISH.
```

- The SET command specifies that procedure output be sent to the default disk file SPSS.LIS and that data results be written on the file NEWDAT.DAT. Both files are written in the current directory.
- The GET command reads the system file INVENT.SYS from the current directory.



- The COMPUTE command modifies the variable TOTCOST.
- The SELECT IF command tells SPSS/PC to select cases in which the value of variable V3 is less than 7.
- The WRITE command writes the values of variable TOTCOST and variables between and including ID and V15 (provided the value of V3 is less than 7). The data are written on the file NEWDAT.DAT.

## Command Printback

SPSS/PC provides two options for printing back commands either on the screen or in output using the INCLUDE and ECHO subcommands.

- If you don't want to see any SPSS/PC commands on your screen, specify INCLUDE=OFF prior to using the INCLUDE command.
- If both SCREEN=ON and ECHO=ON are specified, SPSS/PC echoes the commands from the INCLUDE file to your screen regardless of how INCLUDE is operating. You can set SCREEN=OFF and INCLUDE=OFF to suspend the echoing of commands to the screen.

**INCLUDE** *Printback of commands from command files on the screen.* Specify ON or OFF. INCLUDE=ON (the default) displays commands from the command file on your screen. You can turn off this option by specifying INCLUDE=OFF within the command file. Inline data from command files is not sent to the screen.

**ECHO** *Printback of commands in output.* By default, SPSS/PC sends only results of statistical and reporting procedures to disk and printer files (ECHO=OFF). Specify ECHO=ON to include SPSS/PC commands in disk and printer output.

### Example

```
SET ECHO=ON/DISK=ON.
GET FILE='MUFIL1.SYS'.
CROSSTABS TABLES=AVAR BY BVAR BY CVAR.
SET SCREEN=OFF.
INCLUDE 'REG.INC'.
FIN.
```

- The first SET command integrates commands with output in the default disk file SPSS.LIS.
- The second SET command stops sending output display to the screen. Output is sent to the disk file.
- The INCLUDE command processes commands from file REG.INC in the current directory. By default, all commands from REG.INC are sent to the screen as they are processed. If INCLUDE=OFF is specified in REG.INC, commands are not sent to the screen as they are processed.

## Prompts, Command Terminator, and Sound

Use the PROMPT, CPROMPT, and ENDCMD subcommands to control the characters used for prompts and the command terminator. Use the BEEP subcommand to suppress the beep that signals the next screen of output and errors.

- SPSS/PC uses two types of prompts for command input: command prompt and continuation prompt. The command prompt prompts for a new command and the continuation prompt prompts for the next line of a command.
- The continuation prompt is always issued if a command terminator has not been used on the previous input.
- You can specify special characters for both PROMPT and CPROMPT.
- Every SPSS/PC command must end with a command terminator. SPSS/PC continues to read lines as part of the current command until it encounters the command terminator.
- The ENDCMD specification should be a printable character.
- A high-pitched beep reminds you to page to the next screen of output.
- A low-pitched beep is sounded when an error occurs.

- PROMPT** *The command prompt.* Specify a string of up to 8 characters in apostrophes. The default is SPSS/PC.
- CPROMPT** *The continuation prompt.* Specify a string of up to 8 characters in apostrophes. The default is
- ENDCMD** *The command terminator.* Specify a character in apostrophes. The default is a period.
- BEEP** *Beep.* To suppress the beep, specify OFF. To turn it on, specify ON. The default is ON.

**Example** SET PROMPT='====>' /CPROMPT='this it?'/BEEP=OFF.

- The SET command specifies a new command prompt. The continuation prompt is the question, this it? The BEEP is turned off for this session.

### Treatment of Numeric Blanks

By default, SPSS/PC translates entirely blank fields read with a numeric format to the system-missing value. You can use the BLANKS subcommand to specify some other value.

- BLANKS controls only the translation of numeric fields.
- If a blank field is read with an A format, the resulting value is a blank.
- The BLANKS specification controls all numeric variable blank values. You cannot have different specifications for different variables.
- BLANKS must be specified before data are read.

**BLANKS** *Blanks for numeric fields.* Specify any number. Blanks are translated to this number and the number is not automatically defined as missing. The default is the system-missing value.

**Example** SET BLANKS=-1.

- The SET command translates all numeric-variable blanks to the value -1.

### Random Number Seed

SPSS/PC has a random number generator used by the SAMPLE command and the NORMAL and UNIFORM functions. By default, the seed is a random seed generated from the clock. Optionally, you can specify a seed on the SEED subcommand.

- You can change the random number seed any number of times within a session.
- To replicate samples across sessions or procedures, specify the same seed each time.

**SEED** *The random number seed.* Specify a positive integer or keyword RANDOM. The default is RANDOM.

**Example** SET SEED=200000000.

- The random number seed is set to the value 200000000.

---

# SHOW

SHOW

**Overview** The SHOW command displays a table of all the current specifications on the SET command.

**Syntax** • The minimum specification is simply the command keyword. SHOW has no additional specifications.

**Operations** • SHOW is an operations command and is executed immediately.  
• SHOW lists every current SET specification, including the default settings.

**Limitations** • SHOW BOXSTR and HISTOGRAM graphics characters that are displayed on the screen can be printed on graphics printers but may not be printable on nongraphics printers.

**Example** SHOW.

• The SHOW command produces a table similar to the one below of the current specifications on the SET command. In this example, the settings shown are the default settings.

```
VALUES SET
DISK      OFF      SCREEN  ON       INCLUDE ON
LOG       ON       PRINTER OFF     BEEP    ON
RESULTS                                     EJECT  OFF
PROMPT    SPSS/PC  BOX     -|+      WIDTH   79
CPROMPT                                     BLOCK   ■      LENGTH  24
ENDCMD                                         HIST    ■      SEED    564738991
```

---

## SORT CASES

```
SORT CASES [BY] varlist [{A}] [varlist...]  
                {D}
```

### Example:

```
SORT CASES BY XVAR (A) YVAR (D).
```

**Overview** The SORT CASES procedure reorders the sequence of cases in the active file based on the values of one or more variables.

**Defaults** By default, cases are sorted in ascending order for each variable, starting with the first variable named. For each subsequent variable, cases are sorted in ascending order within categories of previously named variables.

**Tailoring** Optionally, you can sort cases in descending order or use combinations of ascending and descending order for different variables.

**Syntax**

- The minimum specification is a list of variables that are used as sort keys.
- Variables can be numeric or string.
- Keyword BY is optional.
- You can explicitly request the default sort order (ascending) by specifying A in parentheses after the variable name.
- To sort cases in descending order, specify (D).
- An order specification (A or D) applies to all variables to its left that contain no order specification. Thus, if you combine ascending and descending order on the same specification, you may need to specify the default (A) explicitly.

**Operations**

- SORT CASES is a procedure and causes the data to be read.
- SORT CASES begins by sorting the file according to the first variable named. For subsequent variables, cases are sorted within categories of previously named variables.
- For string variables, order specification A implies alphabetical order and D implies reverse alphabetical order.
- The values of system variable \$CASENUM are not changed from their original values as the file is sorted. You can use the variable \$CASENUM as a sorting key.

**Limitations**

- Up to 10 variables can be used as sort keys.
- SORT CASES uses intermediate scratch files for which sufficient disk space must be available.

**Example** SORT CASES BY XVAR (A) YVAR (D).

- Cases are sorted in ascending order of variable XVAR.
- Cases are further sorted in descending order of YVAR within categories of XVAR.

**Example** SORT CAS XVAR YVAR (A) ZVAR (D).

- Cases are sorted in ascending order of XVAR.
- Cases are further sorted in ascending order of YVAR within values of XVAR. Specification A applies to both XVAR and YVAR.
- Cases are further sorted in descending order of ZVAR within values of YVAR and XVAR.
- This example takes advantage of spelling permitted by three-character truncation of commands.

---

## SUBTITLE

```
SUBTITLE [{}]' text [{}]'
```

### Example:

```
SUBTITLE FREQUENCIES FOR SEX=FEMALE..
```

**Overview** The SUBTITLE command inserts a subtitle on the second line from the top of each page of the display file. The default subtitle is a blank line.

- Syntax**
- The only specification on SUBTITLE is the subtitle itself.
  - The subtitle can include any characters.
  - The subtitle can be up to 64 characters long. Subtitles longer than 64 characters are truncated.
  - You can specify the subtitle as a string by enclosing it within either apostrophes or quotation marks (see Universals: Strings). This allows you to include quotation marks, apostrophes, and lower-case text within the subtitle.
  - More than one SUBTITLE command is allowed in a single session.
  - SUBTITLE cannot appear between BEGIN DATA and END DATA.
  - SUBTITLE cannot appear between lines of a multiple-line command.

- Operations**
- Each SUBTITLE command overrides the previous one and takes effect on the next display page.
  - SUBTITLE is independent of TITLE and can be changed separately.

**Example**

```
TITLE 'Preliminary analysis of Santa Survey'.  
SUBTITLE FREQUENCIES FOR SEX=FEMALE..
```

- This subtitle includes a period within the text as well as the required command terminator.

**Example**

```
SUB "Crosstabs of Respondent's Age by Gift Preference":
```

- This subtitle is specified within quotation marks to allow the use of an apostrophe in the subtitle and to preserve upper and lower case.
- This example takes advantage of spelling permitted by three-character truncation of keywords.

---

## TITLE

TITLE [{{' }}] text [{{' }}]

### Example:

```
TITLE This run will contain preliminary analyses.
```

### Overview

The default page heading on SPSS/PC output contains the date, SPSS/PC, and the display page number. The TITLE command replaces the title portion (SPSS/PC) of the default display heading with user-supplied text. The date and page number remain.

### Syntax

- The only specification on TITLE is the title itself.
- The title can include any characters.
- The title can be up to 58 characters long. Titles longer than 58 characters are truncated.
- You can specify the title as a string by enclosing it within either apostrophes or quotation marks (see Universals: Strings). This allows you to include quotation marks, apostrophes, and lower-case text within the title.
- More than one TITLE command is allowed in a single session.
- TITLE cannot appear between lines of a multiple-line command.
- TITLE cannot appear between BEGIN DATA and END DATA.

### Operations

- Each TITLE command overrides the previous one and takes effect on the next display page.
- TITLE does not cause a new page.

### Example

```
TITLE This run will contain preliminary analyses.
```

- This title will replace the default title on subsequent output pages.
- Note that the period (.) is the command terminator and will not appear as part of the title.

### Example

```
TITLE "Regression on Nora's Dissertation Data".
```

- This title is specified within quotation marks to allow the use of an apostrophe in the title and to preserve upper and lower case.

# T-TEST

## Independent Samples:

```
T-TEST GROUPS=varname ( ( {1,2 } ) ) /VARIABLES=varlist  
                        {value  
                        {value,value}
```

```
[/OPTIONS=option numbers]
```

## Paired Samples:

```
T-TEST PAIRS=varlist [WITH varlist] [/[PAIRS=] varlist ...]
```

```
[/OPTIONS=option numbers]
```

## Options:

- 1 Include user-missing values    3 Suppress variable labels
- 2 Exclude missing values listwise    5 Special pairing for WITH

## Examples:

```
T-TEST GROUPS=XVAR(1,3)/VARIABLES=YVAR1 YVAR2 YVAR3...
```

```
T-TEST PAIRS=ZVAR1 ZVAR2 ZVAR3/ZVAR4 ZVAR5.
```

## Overview

T-TEST compares sample means by calculating Student's *t* and displays the two-tailed probability of the difference between the means. Statistics are available for either independent samples (different groups of cases) or paired samples (different variables). Other procedures that compare group means are ANOVA and ONEWAY.

## Defaults

In addition to Student's *t*, degrees of freedom, and two-tailed probabilities, T-TEST produces the mean, standard deviation, standard error, and count for each group or variable. The default display includes the variable label and always uses narrow format. Cases with missing values for the variables that specify a test are excluded from that test.

**Independent-Samples Test.** For independent-samples tests, both pooled- and separate-variance estimates are calculated, along with the *F* value used to test homogeneity of variance and its probability. The two-tailed probability is displayed for the *t* value.

**Paired-Samples Test.** For paired-samples tests, the default output includes the difference between the means, the two-tailed probability level for a test of the difference, the correlation coefficient for the two variables, and the two-tailed probability level for a test of the coefficient.

## Tailoring

**Display Format.** You can suppress the printing of variable labels.

**Statistical Display.** You can control which variables are paired in paired-samples tests. There are no optional statistics. All statistics available are displayed by default.

**Missing Values.** You can include cases with user-missing values in the calculation of the statistics for each test. Alternatively, you can exclude cases with missing values for any of the variables named on the command from all analyses.

## Syntax

- To request both independent- and paired-samples tests, specify GROUPS and VARIABLES first, then PAIRS.
- Subcommands can each be specified only once and must be separated by a slash.

## Operations

- T-TEST is a procedure and causes the data to be read.
- If a GROUPS variable is a long string, only the short-string portion is used to identify groups in the analysis.
- Probability levels are two-tailed.
- The T-TEST display always uses narrow format regardless of the width defined on SET.

- The **BOXSTRING** subcommand controls the characters used in the table display (see **SET**).

### Limitations

- Maximum 1 grouping variable per independent-samples test (the **GROUPS** subcommand).
- Maximum 50 analysis variables per independent-samples test (the **VARIABLES** subcommand).
- Maximum 1 each **GROUPS**, **VARIABLES**, and **OPTIONS** subcommands.
- The maximum number of variables for paired-sample tests is the same as the system limit.

### Example

**T-TEST GROUPS=XVAR(1,3)/VARIABLES=YVAR1 YVAR2 YVAR3.**

- This independent-samples example compares the means of the two groups defined by **XVAR** for variables **YVAR1**, **YVAR2**, and **YVAR3**.

### Example

**T-T PAIRS=ZVAR1 ZVAR2 ZVAR3/ZVAR4 ZVAR5.**

- The first analysis list in this paired-samples example compares the means of **ZVAR1** and **ZVAR2**, **ZVAR1** and **ZVAR3**, and **ZVAR2** and **ZVAR3**. The second analysis list compares the means of **ZVAR4** and **ZVAR5**.
- This example takes advantage of spelling permitted by three-character truncation of keywords.

## GROUPS and VARIABLES Subcommands

Independent samples t-tests are requested with the **GROUPS** and **VARIABLES** subcommands. The **GROUPS** subcommand names a variable used to group cases. The **VARIABLES** subcommand names the dependent variables.

The minimum specification for an independent-samples test is the **GROUPS** subcommand with a variable name followed by the **VARIABLES** subcommand with a variable name.

- **GROUPS** can name only one variable, which can be numeric or string.
- The **GROUPS** variables can be followed by a range specification in parentheses.
- A single **GROUPS** value in parentheses groups all cases with a code equal to or greater than the value into one group and the remaining cases into the other group.
- Two **GROUPS** values in parentheses include cases with the first value in one group and cases with the second value in the other group. Cases with other values are excluded.
- If no **GROUPS** values are specified, (1,2) is assumed for numeric variables and (DEFAULT1, DEFAULT2) is assumed for string variables.
- **VARIABLES** names the variables being analyzed.
- **VARIABLES** can name only numeric variables.

## PAIRS Subcommand

Use the **PAIRS** subcommand to request paired-samples tests.

- The minimum specification for a paired-samples test is the **PAIRS** subcommand with a single analysis list.
- The minimum analysis list is two variables.
- Analysis lists can name only numeric variables.
- An analysis list without keyword **WITH** compares each variable with every other variable.
- An analysis list with keyword **WITH** compares every variable to the left of the keyword with every variable to the right of the keyword.



- Multiple PAIRS subcommands can be used. The keyword PAIRS is required only on the first PAIRS subcommand.
- Use a slash to separate multiple PAIRS subcommands.

### Display Format

The default display includes the variable label. You can change this default with the following option on the OPTIONS subcommand:

**Option 3** *Suppress variable labels.*

### Statistical Display

All statistics available with T-TEST are displayed automatically. However, you can control the pairing of variables in the analysis list with the following option on the OPTIONS subcommand:

**Option 5** *Special pairing for paired-samples test.* Must be used with the keyword WITH. The first variable before WITH is compared to the first variable after WITH, the second variable before WITH is compared to the second variable after WITH, and so forth. The same number of variables should be specified before and after WITH; unmatched variables are ignored. Option 5 is ignored with independent samples or when keyword WITH is not specified in the PAIRS subcommand.

### Example

```
T-TEST PAIRS=SCALE1 TO SCALE3 WITH SCALE4 TO SCALE6
/OPTIONS=5.
```

- This analysis compares SCALE1 and SCALE4, SCALE2 and SCALE5, and SCALE3 and SCALE6.

### Missing Values

For independent-samples tests, cases missing on either the grouping variable or the analysis variable are excluded from the analysis of that variable. For paired-samples tests, a case missing on either of the variables in a given pair is excluded from the analysis of that pair. You alter the handling of cases with missing values by specifying the following on the OPTIONS subcommand:

**Option 1** *Include user-missing values.* Cases with user-missing values are included in the calculation of test statistics.

**Option 2** *Exclude missing values listwise.* A case with a missing value for any variable named on the command is excluded from all analyses the command requests.

## VALUE LABELS

```
VALUE LABELS varlist value 'label' value 'label' ...  
[ /variables ]
```

Example:

```
VALUE LABELS XVAR TO ZVAR 0 'NORTH' 1 'CENTRAL' 2 'WEST'  
3 'SOUTH' 4 'EAST'  
RESIDE 'URBAN' "Within city proper" 'SUBURBAN' "Metro area"  
'EXURBAN' "Outside SMSA".
```

**Overview** The VALUE LABELS command provides descriptive labels for individual values of a variable.

**Syntax**

- The VALUE LABELS command is followed by a variable name (or variable list) and the individual values with associated labels.
- The variable must have been previously defined, either on a DATA LIST, GET, or IMPORT command or on a transformation command.
- Value labels must be specified as strings enclosed in apostrophes or quotation marks (see Universals: Strings).
- Each value label can be up to 20 characters long.
- The values of short string variables as well as their labels are specified as strings.
- You cannot assign value labels to long string variables.
- The same labels can be assigned to the same values of several variables by listing all of the variables followed by the values and associated labels.
- The value labels for one variable or variable list must be separated from the next variable or variable list by a slash.
- More than one VALUE LABELS command is allowed.

**Operations**

- VALUE LABELS is a transformation command and is executed when the data are read for the next procedure.
- Value labels are stored in the active file dictionary.
- If value labels are assigned to any variable that already has value labels assigned to it, the new assignment completely replaces the old assignment.

**Limitations**

- Some procedures print fewer than 20 characters in labels.
- SPSS/PC stores value and variable labels in a fixed area. This area holds approximately 5000 characters.

**Example**

```
VALUE LABELS XVAR TO ZVAR 0 'NORTH' 1 'CENTRAL' 2 'WEST'  
3 'SOUTH' 4 'EAST'  
AVAR 'X' 'SALES STAFF' 'S' 'SUPERVISORY STAFF'  
'M' 'MANAGERIAL STAFF'.
```

- Labels are assigned to the values 0, 1, 2, 3, and 4 of the variables between and including XVAR and ZVAR on the active file.
- Following the required slash, labels for values X, S, and M of variable AVAR are specified. X, S, and M are string values and must be enclosed in apostrophes or quotes.

### Numeric Value Labels

- If the value specified for a numeric variable is shorter than the variable's length, the specified value is right-padded with blanks for matching.
- If the value specified for a numeric variable is longer than the variable's length, the values in the data are right-padded with blanks for matching.

**Example**

```

DAT LIS / ID 1-3 SOCROLE 4-5 OPIN1 TO OPIN5 6-10,
VAL LAB SOCROLE 1 'MOTHER' 2 'FATHER' 3 'SISTER' 4 'BROTHER'
5 'COUSIN' 10.0 'NEIGHBOR' 97 'REFUSED TO ANSWER' 98 "DON'T KNOW"
/OPIN1 TO OPIN5 1 'STRONGLY AGREE' 3 'NO OPINION'
5 'STRONGLY DISAGREE'.
BEG DAT.
0012 12121
0021033333
0039845541
lines of data
END DAT.

```

- The DATA LIST command defines ID as a three-column variable, SOCROLE as a two-column variable, and each of the five OPIN variables as one-column variables.
- The VALUE LABELS command assigns labels to variables SOCROLE, OPIN1, OPIN2, OPIN3, OPIN4, and OPIN5.
- The values associated with labels MOTHER, FATHER, SISTER, BROTHER, and COUSIN are shorter than the width defined for variable SOCROLE. Thus, the specified values are padded for matching. The value associated with label NEIGHBOR is longer than the defined width. VALUE LABELS pads the data value for matching.
- The value label for value 98 is enclosed in quotation marks to allow the embedded apostrophe.
- Three values of the OPIN variables are assigned labels.
- This example takes advantage of spelling permitted by three character truncation of keywords.

### String Value Labels

- Short string values for a variable must contain the same number of characters as the assigned format for that variable. You must include leading and trailing blanks for each string value.

**Example**

```

DATA LIST / REGION 1 RESIDE 2-9 (A) V1 TO V15 10-24.
VAL LAB RESIDE 'URBAN      ' "Within city proper" 'SUBURBAN'
"Metrol area" 'EXURBAN  ' "Outside SMSA"

```

- The DATA LIST command defines one short string variable, RESIDE, which is eight characters wide.
- The VALUE LABELS command assigns labels to three values of RESIDE. Each value is specified in apostrophes, and each label is specified in quotation marks. The value URBAN contains three trailing blanks in order to match the format width of eight characters. The value EXURBAN contains one trailing blank in order to match data values.

---

## VARIABLE LABELS

VARIABLE LABELS varlist 'label' [[/]varname ...]

### Example:

```
VARIABLE LABELS XVAR 'SALES TERRITORY X'  
/YVAR "John's Territory"  
/ZVAR 'SALES TERRITORY Z'.
```

**Overview** The VARIABLE LABELS command assigns a descriptive label of up to 40 characters to a variable.

- Syntax**
- The variable name or list of variables is followed by at least one blank or comma and the associated label.
  - Variable labels must be specified within apostrophes or quotation marks (see Universals: Strings).
  - The variable must have been previously defined, either on a DATA LIST, GET or IMPORT command, or on a transformation command.
  - The slash separating a variable name and its label from subsequent names and labels is optional.
  - More than one VARIABLE LABELS command is allowed per session.
  - Individual variable labels cannot be specified across multiple lines.

- Operations**
- VARIABLE LABELS is a transformation command and is executed when the data are read for the next procedure.
  - If a label is assigned to a variable that already has a label assigned to it, the new label completely replaces the old label.
  - Variable labels are stored in the active file dictionary.
  - The same label is assigned to all variables named before a label specification.

- Limitations**
- Some procedures are unable to show all 40 characters of a label.
  - Variable labels (and value labels) are maintained in a fixed-length area of SPSS/PC. The maximum number of characters for all variable and value labels combined is approximately 5,000.

### Example

```
VARIABLE LABELS XVAR 'SALES TERRITORY X'  
/YVAR "John's Territory"  
/ZVAR 'SALES TERRITORY Z'.
```

- This command assigns labels to variables XVAR, YVAR, and ZVAR.
- The label for YVAR is enclosed in quotation marks to allow the inclusion of an apostrophe in the label.

## WEIGHT

```
WEIGHT {BY varname}
        {OFF}
```

### Example:

```
WEIGHT BY AVAR.
FREQ VAR=VAR1.
```

**Overview** The WEIGHT transformation reassigns the value of the system variable \$WEIGHT for each case (see Universals: System Variables). With WEIGHT, you can arithmetically alter the sample size or its distribution.

**Syntax**

- Only one variable can be specified.
- The WEIGHT variable must be numeric.
- The keyword BY is required to initiate WEIGHT.
- WEIGHT OFF turns weighting off.

**Operations**

- WEIGHT is a transformation and is executed when data are read for the next procedure.
- The default value of \$WEIGHT for each case is 1.0.
- Weighting is permanent during a job unless it is changed or turned off with the WEIGHT OFF specification.
- Each WEIGHT command overrides the previous one.
- WEIGHT tells SPSS/PC to use the value of the specified variable to arithmetically replicate cases for subsequent procedures.
- Weight values do not need to be integer.
- Cases are *not* physically replicated.
- System-missing, user-missing, and negative values for the weighting variable are treated as 0 in the computation of weights.
- A file saved with EXPORT or SAVE when weighting is in effect maintains the weighting in the system variable \$WEIGHT. When the file is reread, weighting is maintained.
- If the weighted number of cases exceeds the sample size, tests of significance are inflated; if it is smaller, they are deflated.

**Example**

```
WEIGHT BY AVAR.
FREQ VAR=VAR1.
```

- The frequency counts for the values of variable VAR1 will be weighted (multiplied) by the values of variable AVAR.

```
COMPUTE WVAR=1.
IF (AVAR EQ 1000) WVAR=.5.
WEIGHT BY WVAR.
```

**Example**

- Variable WVAR is initialized to 1 with the COMPUTE command. The value of WVAR is changed to .5 with the IF command for cases where AVAR equals 1000.
- Subsequent procedures will use a case base derived from the arithmetic replication of cases.

## WRITE

```
WRITE {VARIABLES={ALL** }  
      {varlist}  
  
      [/CASES={FROM {1**} } [TO {eof**} ]  
            {n}          {n}  
  
            [BY {1**} ] ]  
            {n}  
  
            [/FORMAT={ {UNNUMBERED**} } { {WRAP**} }  
                   {NUMBERED}         {SINGLE}  
  
            [WEIGHT]]
```

\*\*Default if subcommand is omitted.

### Example:

```
WRITE VARIABLES=XVAR AVAR /CASES=10 TO 100 BY 2.
```

- Overview** The WRITE procedure writes cases to an ASCII file. The cases are written to the results file specified on SET (by default, SPSS.PRC). SPSS/PC automatically specifies the format of the case file that is written using the dictionary formats and a record length of 80. In addition to writing the file, WRITE displays a table that shows the names and the record and column locations of the variables on the written file.
- Defaults** By default, all user-defined variables are written for all cases in the file. Each case is written on as many records as are needed.
- Tailoring** You can limit the written file to specified variables or a specified sequence of cases. In addition, you can include each case's weight or sequence number in the file. You can also limit each case to a single record.
- Syntax**
- The minimum specification is simply the command keyword.
  - All subcommands are optional.
  - Subcommands can be specified in any order and must be separated by slashes.
- Operations**
- WRITE is a procedure and causes the data to be read.
  - WRITE uses the dictionary print formats (see Universals: Formats). If a value is greater than the format allows, asterisks (\*) are written.
  - Records are written with 80 characters regardless of the width defined on SET.
  - If a long string variable cannot be written within the record length of 80, it is truncated.
  - Records are always written with a single blank between variables.
  - System-missing values are written as a period (.).
  - System variables are only written when specifically requested.
  - Records are written to the results file named on SET. The default is SPSS.PRC.
  - If you write to the same file with more than one procedure in the same session, the file will contain only the results of the last procedure. Use the SET command to control the results file you write to if you write output from more than one procedure.
- Limitations**
- The VARIABLES, FORMAT, and CASES subcommands can each be specified only once.

### Example

```
SET SCREEN=ON/DISK=ON.  
DATA LIST FILE='EMPLOYEE.DAT'  
/XVAR 1 YVAR 10-15 ZVAR 3-9(2) /3 AVAR 25-30(A).  
WRITE VARIABLES=XVAR AVAR /CASES=10 TO 100 BY 2.
```

- This example writes a file containing every second case starting with Case 10 and stopping at 100 for variables XVAR and AVAR.
- The records are written to the default results file SPSS.PRC.
- The display table describing the format of the written file is sent to the screen and to the default disk file SPSS.LIS.

### VARIABLES Subcommand

By default, all user-defined variables in the active file are included. Use the optional VARIABLES subcommand to control the variables to be written for each case.

- If you use the VARIABLES subcommand, a list of variables must be specified and only specified variables are written.
- Variables named on VARIABLES must already exist.
- Variables are written in the order named on VARIABLES.
- If a variable is named more than once it is written more than once.

Optionally, you can specify keyword ALL on the VARIABLES subcommand:

**ALL** Write all user-defined variables. Variables are written in the order in which they appear in the active file. This is the default if the VARIABLES subcommand is omitted.

### CASES Subcommand

By default, all cases in the active file are written. Use the CASES subcommand to limit the number and pattern of cases written.

- If you omit CASES or use it without specifications, all cases in the active file are written.
- Only defaults that you change are altered.

The following can be specified on CASES:

**FROM n** The case number of the first case to be written. The default is 1.

**TO n** Upper limit on the cases to be written. The default is the end of the file. CASES 100 is interpreted as CASES TO 100.

**BY n** Increment used to choose cases for writing. The default is 1.

### FORMAT Subcommand

By default, the written file does not include the case weight or sequence number for each case, and each case uses as many records as it needs. Use the FORMAT subcommand to specify format options for the written file.

- If the FORMAT subcommand is omitted or if it is included without specifications, the file is written with the default format.
- Only defaults that you change are altered.

The following may be specified on FORMAT:

**NUMBERED** Include the sequence number of the case. The sequence number is the first item written for each case and is labeled Case# in the display table. The default is UNNUMBERED.

**SINGLE** Limit each case to one record. If variables requested cannot be written on a single record, WRITE is not executed. The default is to WRAP the case, using as many records as necessary.

**WEIGHT** Write the value of the case's weight in the active file.